Supplementary material

S1 Parameter estimates with relative standard errors (RSE) of the final model excluding one strongly deviating subject.

```
; Model description: Study ACTG 5267: BDQ + M2 + M3

; Elin Svensson, Pharmacometrics Research Group, Uppsala University


$PROB Study ACTG5267

$DATA TMC207_EFV_ACTGstudy20120717.csv IGNORE=@

$INPUT ID DUMID DROP STUDYDAY DAT2=DROP TIME TAD TMCDOSE DOSE2 EVID MDV AMT DROP DROP FLAG CMT DVMG LNDVMG DVMOL
CPAR CM2 DV L2 ZEROO LLOQ BQL EFV GENDER RACE DROP AGE DROP DROP METAB HT WTOLD WT ALB


$SUBROUTINE ADVAN6 TOL=6


$MODEL NCOMPARTMENTS=8

COMP=(DEPOT DEFDOSE)

COMP=(CENTRAL DEFOBSERVATION)

COMP=(PERI1)

COMP=(PERI2)

COMP=(M2)

COMP=(M3)

COMP=(M2P1)
```

```
COMP=(M3P1)


$PK
;--- Typical values fixed effects
TVMTT =         THETA(1)
TVKA=           THETA(2)
TVCL=           THETA(3)
TVV=            THETA(4)
TVQ1 =          THETA(5)
TVVP1 =         THETA(6)
TVQ2 =          THETA(7)
TVVP2 =         THETA(8)
TVCLM2 =        THETA(9)
TVVM2 =         THETA(10)
TVQ1M2 =        THETA(11)
TVVP1M2 =       THETA(12)
EFVEFF1 =       THETA(13)
TVCLM3=         THETA(14)
TVVM3=          THETA(15)
TVQ1M3=         THETA(16)
```

```
TVVP1M3=        THETA(17)
EFVEFFM3 =      THETA(18)



;--- Variability
BOVF =                   ETA(1)
IF(TMCDOSE.EQ.2) BOVF = ETA(2)
BSVF =                   ETA(3)
BOVMTT =                 ETA(4)
IF(TMCDOSE.EQ.2) BOVMTT =    ETA(5)
BSVCL=                   ETA(6)
BSVCLM2=                 ETA(7)
BSVCLM3=                 ETA(8)
BSVEFVEFF1 =             ETA(9)
BSVEFVEFF2 =             ETA(10)
BSVEFVEFF3 =             ETA(11)
BSVV=                    ETA(12)
BSVQ1=                   ETA(13)
BSVVM2=                  ETA(14)
BSVVP1M2=                ETA(15)
```

```
; Allomertic scaling
ALLCL = (WT/70)**0.75
ALLV= WT/70

; Effect of EFV on CL
EFFE1 = EFVEFF1*EXP(BSVEFVEFF1)
EFFE2 = EFVEFF1*EXP(BSVEFVEFF2)
EFFE3 = EFVEFFM3*EXP(BSVEFVEFF3)

;--- Parameters
KA =            TVKA
MTT =           TVMTT*EXP(BOVMTT)
CL =            TVCL*ALLCL*EXP(BSVCL)
IF(TAD.GT.504.OR.TMCDOSE.EQ.2)  CL =  TVCL*ALLCL*EXP(BSVCL)*EFFE1
V =             TVV*ALLV*EXP(BSVV)
Q1=             TVQ1*ALLCL*EXP(BSVQ1)
VP1 =           TVVP1*ALLV
Q2 =            TVQ2*ALLCL
VP2=            TVVP2*ALLV
```

```
CLM2 =          TVCLM2*ALLCL*EXP(BSVCLM2)
IF(TAD.GT.504.OR.TMCDOSE.EQ.2) CLM2 = TVCLM2*ALLCL*EXP(BSVCLM2)*EFFE2
VM2=            TVVM2*ALLV*EXP(BSVVM2)
Q1M2 =          TVQ1M2*ALLCL
VP1M2=          TVVP1M2*ALLV*EXP(BSVVP1M2)
CLM3 =          TVCLM3*ALLCL*EXP(BSVCLM3)
IF(TAD.GT.504.OR.TMCDOSE.EQ.2) CLM3 = TVCLM3*ALLCL*EXP(BSVCLM3)*EFFE3
VM3 =           TVVM3*ALLV
Q1M3 =          TVQ1M3*ALLCL
VP1M3 =         TVVP1M3*ALLV


;--- Scaling factors
S2=V
S5=VM2
S6=VM3

;--- To stratify VPC
STRT = EFV+2*FLAG
```

```
;--- Reorder to get correct order in VPC
VPCLOC = 1
IF(STRT.EQ.2) VPCLOC= 4
IF(STRT.EQ.4) VPCLOC= 5
IF(STRT.EQ.5) VPCLOC= 2
IF(STRT.EQ.6) VPCLOC= 6
IF(STRT.EQ.7) VPCLOC= 3

;--- Calc time after dose
TADD = TAD/24
TADW = TAD/(24*7)

;--- Transit model - code by Paolo Denti

IF (NEWIND.NE.2.OR.EVID.GE.3) THEN ; beginning of dataset, or new individual
        TNXD=TIME ; TIME will be the time of the first record of that subject even if it's not a dose, but...
        PNXD=AMT ; ...the amount will be 0 if the first record is not a dose, so no problem.
ENDIF

TDOS=TNXD ; This will either save here the temporary values if it's a new individual...
```

```
PD=PNXD ; ...or the values which were read one record ahead during the execution of the previous record.


IF(AMT.GT.0) THEN ; This reads one record ahead and stores the data to be used when running the following record
        TNXD=TIME
        PNXD=AMT
ENDIF


IF (DOSTIM.GT.0) THEN ; This will account for the ADDL or lagged doses
        TNXD=DOSTIM
        PNXD=AMT
ENDIF


F1=0            ; Nothing goes directly in to compartment 1
BIO=720.072*EXP(BOVF+BSVF) ; AMT=1 in input file, dose was 400 mg, MW TMC207 555.5 g/mol, DV as nmol/mL = µmol/L -->
(400/1000)/(555.5)*1000000 = 720.1 µmol


NN=THETA(19)
KTR=((NN+1)/MTT)
L      = 0.9189385 + (NN + 0.5)*LOG(NN) - NN + LOG(1 + 1/(12*NN)) ; approximation to the natural logarithm of the gamma function
```

```
;--- Shorter run time when calculations outside $des block
BIOPD=BIO*PD
IF (BIOPD.EQ.0) BIOPD=BIOPD+0.00001
LBPD  = LOG(BIOPD)
LKTR  = LOG(KTR)
PIZZA = LBPD + LKTR - L



$DES
TEMPO=T-TDOS ; This is time after dose, it should always be >= 0
IF(TEMPO.GT.0) THEN
        KTT=KTR*TEMPO
        DADT(1) = EXP(PIZZA+NN*LOG(KTT)-KTT) -A(1)*KA
ELSE
        DADT(1)=0 ; Executed only when TEMPO=0, or before the first dose is given
ENDIF
DADT(2) = A(1)*KA - A(2)*CL/V - A(2)*Q1/V + A(3)*Q1/VP1 - A(2)*Q2/V + A(4)*Q2/VP2          ; BDQ
DADT(3) = A(2)*Q1/V - A(3)*Q1/VP1
DADT(4) = A(2)*Q2/V - A(4)*Q2/VP2
DADT(5) = A(2)*CL/V - A(5)*CLM2/VM2 - A(5)*Q1M2/VM2 + A(7)*Q1M2/VP1M2          ; M2
```

```
DADT(6) = A(5)*CLM2/VM2 - A(6)*CLM3/VM3    - A(6)*Q1M3/VM3 + A(8)*Q1M3/VP1M3          ; M3

DADT(7) = A(5)*Q1M2/VM2 - A(7)*Q1M2/VP1M2

DADT(8) = A(6)*Q1M3/VM3 - A(8)*Q1M3/VP1M3


$ERROR

DEL= 1E-12

IPRED=LOG(F+DEL) ; DV log-transformed


W = 1

IF(TAD.LE.6) W= W* THETA(20)             ; Weighting of residual error of samples during first 6 hours

BQLW=1

IF(BQL.EQ.1) BQLW=THETA(21)                      ;  Weighting of residual error of samples < BQL


IRES=DV-IPRED

IWRES=IRES/W


Y = IPRED + W*BQLW*EPS(1)                                ; BDQ

IF(FLAG.EQ.2) Y = IPRED + W*BQLW*EPS(2)                  ; M2

IF(FLAG.EQ.3) Y = IPRED + W*BQLW*EPS(3)                  ; M3
```

$THETA

(1E-06,1.31259) ; 1 MTT

(1E-06,0.127908) ; 2 KA

(1E-06,2.96031) ; 3 CL

(1E-06,17.3388) ; 4 V

(1E-06,5.01325) ; 5 Q1

(1E-06,2871.59) ; 6 VP1

(1E-06,4.15909) ; 7 Q2

(1E-06,135.807) ; 8 VP2

(1E-06,12.3361) ; 9 CLM2

(1E-06,658.569) ; 10 VM2

(1E-06,102.953) ; 11 Q1M2

(1E-06,2836.17) ; 12 VP1M2

(1,2.06878) ; 13 EFVEFF BDQCL

(1E-06,39.1751) ; 14 CLM3

(3,11.2449) ; 15 VM3

(1E-06,106.107) ; 16 QM3

(1E-06,2679.86) ; 17 VPM3

(1,1.12233) ; 18 EFVEFF M3

```
 (0,5.20689) ; 19 NN
 (0,1.8709) ; 20 Weigting of samples <6h
 (0,3.28215) ; 21 Weighting of BQL samples

$OMEGA  BLOCK(1)
 0.0555944  ;  1   BOV F
$OMEGA  BLOCK(1) SAME
$OMEGA
 0.058862  ;  3   BSV F
$OMEGA  BLOCK(1)
 0.3074  ; 4   BOV MTT
$OMEGA  BLOCK(1) SAME
$OMEGA  BLOCK(6)
 0.0559648  ; 6   BSV CLBDQ
 0.0131626 0.035439  ; 7 BSV CLM2
 -0.00914441 0.0402624 0.0902629  ; 8  BSV CLM3
 -0.0337552 -0.0217602 -0.0138624 0.0425689  ; 9 BSV EFVEFF1
 -0.0198463 -0.0408591 -0.0609976 0.043842 0.0796311  ; 10 BSV EFVEFF2
 -0.00571503 -0.0358474 -0.0820472 0.0286691 0.0811359 0.106536  ; 11 BSV EFVEFF3
$OMEGA
```

```
 0.120173  ; 12   BSV V
 0.034759  ; 13  BSV Q1
 0.0833384  ; 14  BSV VM2
 0.0671284  ; 15  BSV VP1M2

$SIGMA  BLOCK(3)
 0.0572037  ; 1 Prop error BDQ
 0.0220711 0.03124  ; 2 Prop error M2
 0.005659 0.013683 0.0225068  ; 3 Prop error M3

$ESTIMATION METHOD=1 MAXEVAL=9999 INTERACTION PRINT=1 SIGL=6 NSIG=2 PRINT=1 NOABORT MSFO=run569.msf
$COVARIANCE PRINT=E
;$TABLE
```

Supplementary material

S2 Parameter estimates with relative standard errors (RSE) of the final model excluding one strongly deviating subject.

Fixed effects[‡] (RSE)		Random effects [CV%] (RSE)						
MTT [h]	1.38 (14.1%)	BOV F	16.2% (21.1%)					
KA [h^{-1}]	0.127 (9.1%)	BSV F	17.8% (19.0%)					
CL [L h^{-1}]	2.69 (19.7%)	BOV MTT	52.9% (11.2%)					
V [L]	15.6 (18.1%)	BSV CL	24% (11.9%)					
Q1 [L h^{-1}]	4.92 (12.1%)	BSV CLM2	28.1%* (31.5%)	18% (23.7%)				
VP1 [L]	3080 (31.3%)	BSV CLM3	-11.5%* (90.8%)	73.6%* (34.5%)	30.4% (24.9%)			
Q2 [L h^{-1}]	3.98 (9.6%)	BSV EFVEFF-BDQ	-67.3%* (13.8%)	-51.9%* (24.1%)	-23.3%* (64.7%)	20% (10.8%)		
VP2 [L]	135 (12.4%)	BSV EFVEFF-M2	-25.5%* (36.6%)	-73%* (29.5%)	-72.6%* (32.8%)	73.5%* (17%)	27.3% (22.1%)	
CLM2 [L h^{-1}]	11.2 (18.1%)	BSV EFVEFF-M3	-6.4%* (164%)	-58.4%* (43.2%)	-82.8%* (30.3%)	44%* (35.2%)	90.1%* (27.2%)	32.8% (25.0%)
VM2 [L]	620 (12.9%)	BSV V	31.4% (34.8%)					
Q1M2 [L h^{-1}]	95.3 (12.5%)	BSV Q1	18.7% (16.1%)					
VP1M2 [L]	2720 (16.2%)	BSV VM2	26.8% (18.8%)					
CLM3 [L h^{-1}]	2.13 (14%)	BSV VP1M2	15.9% (24.7%)					
VM3 [L]	35.8 (27.2%)							
QM3 [L h^{-1}]	8.38 (1330%)	Prop error TMC	24.2% (5.3%)					
VPM3 [L]	98.4 (52.6%)	Prop error M2	15%* (6.1%)	17.7% (4.8%)				
EFVEFF BDQ and M2	2410 (12%)	Prop error M3	7.5%* (21.4%)	11.3%* (5.6%)	17.7% (10.2%)			
EFVEFF M3	1.14 (21.8%)							
NN	5.12 (24.4%)							
Error weight TAD <6h	1.88 (10.9%)							
Error weight <BQL	3.36 (18.3%)							

Abbreviations: MTT = mean transit time, KA = absorption rate constant, F = bioavailability, CL = clearance, V = volume of distribution, Q = intercompartmental clearance, VP = volume of distribution of peripheral compartments, EFVEFF = induction effect of EFV, NN = number of transit compartmentsTAD = time after dose, BLQ = below limit of quantification, Prop = proportional, BOV = between occasion variability, BSV = between subject variability, RSE = relative standard error, CV% = coefficient of variation as percentage

‡ estimated with typical value of F fixed to one.

* correlation estimated as a covariance.